

Глава 12. Выборка из диапазона дат с помощью критерия в ином формате

Эта глава из книги: Майкл Гирвин. Ctrl+Shift+Enter. Освоение формул массива в Excel.

[Предыдущая глава](#)

[Оглавление](#)

[Следующая глава](#)

При создании формул массива, операторы массива могут значительно увеличить время расчета. В этой короткой главе рассматриваются два примера, в которых обычная формула справляется с задачей значительно быстрее формулы массива.

Подсчет дат, когда критерий сформулирован в виде текста. На рис. 12.1 показан набор данных с датами в стандартном формате Excel, то есть в виде порядковых чисел. В тоже время, критерии заданы как число (год) и текст (месяц). Цель – подсчитать, сколько дат соответствуют критерию. Проблема в том, что у нас несоответствие формата данных: в столбце А даты как порядковые номера, а критерий – смесь чисел и текста. На рис. 12.1 приведено пять различных формул, которые можно использовать для достижения цели.

	А	В	С	Д	Е	Ф	Г	Н	И
1	Дата	Продажи	Вспомогательная						
2	27.10.12	\$5 012	102012	=МЕСЯЦ(А2)&ГОД(А2)					
3	05.10.12	\$6 523	102012						
4	28.12.11	\$1 589	122011						
5	30.10.12	\$2 489	102012						
6	25.10.11	\$5 789	102011						
7	26.11.11	\$5 790	112011						
8	27.11.11	\$5 791	112011						
10	Счет								
11	3	[1]	=СЧЁТЕСЛИ(С2:С8;МЕСЯЦ(F8&1)&E8)						
12	3	[2]	=СЧЁТЕСЛИМН(А2:А8;">=1"&F8&E8;А2:А8;"<="&КОНМЕСЯЦА(1&F8&E8;0))						
13	3	[3]	=СЧЁТЕСЛИ(А2:А8;"<="&КОНМЕСЯЦА(1&F8&E8;0))-СЧЁТЕСЛИ(А2:А8;"<"&1&F8&E8)						
14	3	[4]	=СУММПРОИЗВ(--(МЕСЯЦ(А2:А8)&ГОД(А2:А8)=МЕСЯЦ(1&F8)&E8))						
15	3	[5]	=СУММПРОИЗВ(--(ТЕКСТ(А2:А8;"МММГГГ")=F8&E8))						

Рис. 12.1. Подсчет количества дат (заданных порядковыми номерами) по двум критериям: году (число) и месяцу (текст)

Давайте подробнее изучим работу этих пяти формул.

Формула [1]:

- Если вы можете позволить себе вспомогательный столбец, функция СЧЁТЕСЛИ будет самым простым решением.
- Функция МЕСЯЦ возвращает число между 1 и 12, а функция ГОД – число (год).
- Хотя Excel требует, чтобы аргумент функции МЕСЯЦ был представлен датой в числовом формате, этот аргумент может распознать и текст. Однако МЕСЯЦ(Окт) вернет ошибку, а вот если добавить к названию месяца любое число, например, 1, то Excel справится. Используйте, как в формуле выражение Окт1, заданное фрагментом F8&1, или 1Окт, заданное фрагментом 1&F8.
- Формулы с вспомогательными столбцами как правило работают быстрее.

Формула [2]:

- Если у вас Excel 2007 или более поздний, вы можете использовать функции СЧЁТЕСЛИМН и КОНМЕСЯЦА.
- Вам даны год (в виде числа) и месяц (как текст). Это означает, что вы можете вычислить дату начала и конца месяца, а затем определить даты, попадающие между ними.
- Месяц всегда начинается с первого числа, так что вы можете создать нижнюю границу диапазона конкатенацией: ">=1"&F8&E8. Операции конкатенации возвращают текст, но это не страшно, т.к. функция СУММЕСЛИМН понимает даты в виде текста.
- Вы используете функцию КОНМЕСЯЦА с аргументом *число_месяцев* равным нулю; это позволяет получить последнюю дату текущего месяца. Функция КОНМЕСЯЦА является динамической: она возвращает 28 или 29 для февраля и 30 или 31 для любого другого месяца.
- Эта формула является самой быстрой, если вам нужно получить решение в одной ячейке.

Формула [3]:

- Если у вас Excel версии младше 2007 г., вы можете использовать две функции СЧЁТЕСЛИ, одну – для верхнего диапазона, вторую – для нижнего. Фокус в том, чтобы сначала сосчитать все значения, которые равны или меньше верхней границы, а затем вычесть все значения, которые меньше нижней границы.
- В Excel 2003 или более ранней, чтобы добавить функцию КОНМЕСЯЦА, вам нужно выбрать *Инструменты* → *Надстройки* → *Анализ Данных*.
- Эта формула работает быстрее, чем формулы [4] и [5].

Формула [4]:

- Функции МЕСЯЦ и ГОД возвращают числа, извлекая их из порядкового номера даты.
- Далее сравниваются два фрагмента, каждый полкленный конкатенацией.

Формула [5]:

- Функция ТЕКСТ используется для представления чисел в виде текста. Второй аргумент этой функции – *формат* – определяет, как будет представлено число. Вы можете конвертировать весь столбец А в текст, состоящий из 7 символов: 3 буквы месяца и 4 цифры года.

Нахождение объема продаж за год. На рис. 12.4 показан пример несоответствие формата года в критерии Е6 (число) и формата дат в диапазоне А2:А6 (порядковый номер). Цель – найти сумму продаж за год. На рисунке представлены шесть вариантов формул, которые могут решить задачу. Обратите внимание, что в формулах [1] и [2] критерии начала и конца года жестко зашиты в коде, т.к. они не могут изменяться. Это 1/1 и 31/12). Формулы размещены на рисунке в порядке увеличения скорости работы.

	А	В	С	Д	Е	Ф
1	Дата	Продажи	Вспомогательная			
2	27.10.2012	306	2012	=ГОД(А2)		
3	05.02.2011	530	2011			
4	28.06.2010	167	2010			
5	30.10.2012	639	2012		Критерий:	
6	25.11.2010	635	2010		2012	
8	Сумма					
9	945	[1] =СУММЕСЛИМН(В2:В6;А2:А6;">=1/1/"&Е6;А2:А6;"<=31/12/"&Е6)				
10	945	[2] =СУММЕСЛИ(А2:А6;"<=31/12/"&Е6;В2:В6)-СУММЕСЛИ(А2:А6;"<1/1/"&Е6;В2:В6)				
11	945	[3] =СУММЕСЛИМН(В2:В6;С2:С6;Е6)				
12	945	[4] =СУММЕСЛИ(С2:С6;Е6;В2:В6)				
13	945	[5] =СУММПРОИЗВ(--(ГОД(А2:А6)=Е6);В2:В6)				
14	945	[6] {=СУММ(ЕСЛИ(ГОД(А2:А6)=Е6;В2:В6))}				

Рис. 12.4. Формата года в критерии Е6 (число) не соответствует формату дат в диапазоне А2:А6 (порядковый номер)